NAG Fortran Library Routine Document

F02GCF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F02GCF computes selected eigenvalues and eigenvectors of a complex general matrix.

2 Specification

```
      SUBROUTINE F02GCF(CRIT, N, A, LDA, WL, WU, MEST, M, W, V, LDV, WORK,
     1                  LWORK, RWORK, IWORK, BWORK, IFAIL)
      INTEGER           N, LDA, MEST, M, LDV, LWORK, IWORK(*), IFAIL
      real              WL, WU, RWORK(*)
      complex           A(LDA,*), W(*), V(LDV,MEST), WORK(LWORK)
      LOGICAL           BWORK(*)
      CHARACTER*1       CRIT
```

3 Description

This routine computes selected eigenvalues and the corresponding right eigenvectors of a complex general matrix A:

$$Ax_i = \lambda_i x_i.$$

Eigenvalues λ_i may be selected either by *modulus*, satisfying

$$w_l \leq |\lambda_i| \leq w_u,$$

or by *real part*, satisfying

$$w_l \leq \mathrm{Re}(\lambda_i) \leq w_u.$$

4 References

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: CRIT – CHARACTER*1 *Input*

On entry: indicates the criterion for selecting eigenvalues:

if CRIT = 'M', then eigenvalues are selected according to their moduli: $w_l \leq |\lambda_i| \leq w_u$;

if CRIT = 'R', then eigenvalues are selected according to their real parts: $w_l \leq \mathrm{Re}(\lambda_i) \leq w_u$.

Constraint: CRIT = 'M' or 'R'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

3: A(LDA,*) – ***complex*** array *Input/Output*

Note: the second dimension of the array A must be at least max(1, N).

On entry: the n by n general matrix A.

On exit: A contains the Hessenberg form of the balanced input matrix A' (see Section 8).

4: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F02GCF is called.

Constraint: LDA $\geq$ max(1, N).

5: WL – ***real*** *Input*
6: WU – ***real*** *Input*

On entry: w_l and w_u, the lower and upper bounds on the criterion for the selected eigenvalues (see CRIT).

Constraint: WU > WL.

7: MEST – INTEGER *Input*

On entry: the second dimension of the array V as declared in the (sub)program from which F02GCF is called. MEST must be an upper bound on m, the number of eigenvalues and eigenvectors selected. No eigenvectors are computed if MEST < m.

Constraint: MEST $\geq$ max(1, m).

8: M – INTEGER *Output*

On exit: m, the number of eigenvalues actually selected.

9: W(*) – ***complex*** array *Output*

Note: the dimension of the array W must be at least max(1, N).

On exit: the first M elements of W hold the selected eigenvalues; elements M+1 to N contain the other eigenvalues.

10: V(LDV,MEST) – ***complex*** array *Output*

On exit: V contains the selected eigenvectors, with the ith column holding the eigenvector associated with the eigenvalue λ_i (stored in W(i)).

11: LDV – INTEGER *Input*

On entry: the first dimension of the array V as declared in the (sub)program from which F02GCF is called.

Constraint: LDV $\geq$ max(1, N).

12: WORK(LWORK) – ***complex*** array *Workspace*
13: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F02GCF is called.

Constraint: LWORK $\geq$ max(1, N $\times$ (N + 2)).

14: RWORK(*) – ***real*** array *Workspace*

Note: the dimension of the array RWORK must be at least max(1, 2$\times$N).

15: IWORK(*) – INTEGER array *Workspace*

Note: the dimension of the array IWORK must be at least max(1, N).

16: BWORK(*) – LOGICAL array *Workspace*

Note: the dimension of the array BWORK must be at least max(1, N).

17: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, −1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value −1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value −1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, CRIT ≠ 'M' or 'R',
or N < 0,
or LDA < max(1, N),
or WU ≤ WL,
or MEST < 1,
or LDV < max(1, N),
or LWORK < max(1, N×(N + 2)).

IFAIL = 2

The QR algorithm failed to compute all the eigenvalues. No eigenvectors have been computed.

IFAIL = 3

There are more than MEST eigenvalues in the specified range. The actual number of eigenvalues in the range is returned in M. No eigenvectors have been computed. Rerun with the second dimension of V = MEST ≥ M.

IFAIL = 4

Inverse iteration failed to compute all the specified eigenvectors. If an eigenvector failed to converge, the corresponding column of V is set to zero.

7 Accuracy

If λ_i is an exact eigenvalue, and $\tilde{\lambda}_i$ is the corresponding computed value, then

$$|\tilde{\lambda}_i - \lambda_i| \leq \frac{c(n)\epsilon\|A'\|_2}{s_i},$$

where $c(n)$ is a modestly increasing function of n, ϵ is the ***machine precision***, and s_i is the reciprocal condition number of λ_i; A' is the balanced form of the original matrix A (see Section 8), and $\|A'\| \leq \|A\|$.

If x_i is the corresponding exact eigenvector, and $\tilde{x}_i$ is the corresponding computed eigenvector, then the angle $\theta(\tilde{x}_i, x_i)$ between them is bounded as follows:

$$\theta(\tilde{x}_i, x_i) \leq \frac{c(n)\epsilon\|A'\|_2}{sep_i}$$

where sep_i is the reciprocal condition number of x_i.

The condition numbers s_i and sep_i may be computed from the Hessenberg form of the balanced matrix A' which is returned in the array A. This requires calling F08PSF (CHSEQR/ZHSEQR) with JOB = 'S' to compute the Schur form of A', followed by F08QYF (CTRSNA/ZTRSNA).

8 Further Comments

The routine calls routines from LAPACK in Chapter F08. It first balances the matrix, using a diagonal similarity transformation to reduce its norm; and then reduces the balanced matrix A' to upper Hessenberg form H, using a unitary similarity transformation: $A' = QHQ^H$. The routine uses the Hessenberg QR algorithm to compute all the eigenvalues of H, which are the same as the eigenvalues of A. It computes the eigenvectors of H which correspond to the selected eigenvalues, using inverse iteration. It premultiplies the eigenvectors by Q to form the eigenvectors of A'; and finally transforms the eigenvectors to those of the original matrix A.

Each eigenvector x is normalized so that $\|x\|_2 = 1$, and the element of largest absolute value is real and positive.

The inverse iteration routine may make a small perturbation to the real parts of close eigenvalues, and this may shift their moduli just outside the specified bounds. If you are relying on eigenvalues being within the bounds, you should test them on return from F02GCF.

The time taken by the routine is approximately proportional to n^3.

The routine can be used to compute *all* eigenvalues and eigenvectors, by setting WL large and negative, and WU large and positive. In some circumstances it may do this more efficiently than F02GBF, but this depends on the machine, the size of the problem, and the distribution of eigenvalues.

9 Example

To compute those eigenvalues of the matrix A which lie in the range $[-5.5, +5.5]$, and their corresponding eigenvectors, where

$$A = \begin{pmatrix} -3.97 - 5.04i & -4.11 + 3.70i & -0.34 + 1.01i & 1.29 - 0.86i \\ 0.34 - 1.50i & 1.52 - 0.43i & 1.88 - 5.38i & 3.36 + 0.65i \\ 3.31 - 3.85i & 2.50 + 3.45i & 0.88 - 1.08i & 0.64 - 1.48i \\ -1.10 + 0.82i & 1.81 - 1.59i & 3.25 + 1.33i & 1.57 - 3.44i \end{pmatrix}.$$

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F02GCF Example Program Text
*     Mark 17 Release. NAG Copyright 1995.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, MMAX, LDA, LDV, LWORK
      PARAMETER        (NMAX=8,MMAX=3,LDA=NMAX,LDV=NMAX,LWORK=64*NMAX)
*     .. Local Scalars ..
      real             WL, WU
      INTEGER          I, IFAIL, J, M, N
*     .. Local Arrays ..
      complex          A(LDA,NMAX), V(LDV,NMAX), W(NMAX), WORK(LWORK)
      real             RWORK(2*NMAX)
      INTEGER          IWORK(NMAX)
      LOGICAL          BWORK(NMAX)
```

```
      CHARACTER         CLABS(1), RLABS(1)
*     .. External Subroutines ..
      EXTERNAL          F02GCF, X04DBF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F02GCF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, WL, WU
      IF (N.LE.NMAX) THEN
*
*        Read A from data file
*
         READ (NIN,*) ((A(I,J),J=1,N),I=1,N)
*
*        Compute selected eigenvalues and eigenvectors of A
*
         IFAIL = 0
*
         CALL F02GCF('Moduli',N,A,LDA,WL,WU,MMAX,M,W,V,LDV,WORK,LWORK,
     +               RWORK,IWORK,BWORK,IFAIL)
*
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'Eigenvalues'
         WRITE (NOUT,99999) (W(I),I=1,M)
         WRITE (NOUT,*)
*
         CALL X04DBF('General',' ',N,M,V,LDV,'Bracketed','F7.4',
     +               'Eigenvectors','Integer',RLABS,'Integer',CLABS,80,
     +               0,IFAIL)
*
      END IF
      STOP
*
99999 FORMAT ((3X,4(' (',F7.4,',',F7.4,')',:)))
      END
```

9.2 Program Data

```
F02GCF Example Program Data
  4  -5.5  5.5                                                :Values of N, WL, WU
 (-3.97,-5.04) (-4.11, 3.70) (-0.34, 1.01) ( 1.29,-0.86)
 ( 0.34,-1.50) ( 1.52,-0.43) ( 1.88,-5.38) ( 3.36, 0.65)
 ( 3.31,-3.85) ( 2.50, 3.45) ( 0.88,-1.08) ( 0.64,-1.48)
 (-1.10, 0.82) ( 1.81,-1.59) ( 3.25, 1.33) ( 1.57,-3.44)   :End of matrix A
```

9.3 Program Results

```
 F02GCF Example Program Results

 Eigenvalues
    (-5.0000, 2.0060) ( 3.0023,-3.9998)

 Eigenvectors
                    1                 2
 1  (-0.3865, 0.1732) (-0.0356,-0.1782)
 2  (-0.3539, 0.4529) ( 0.1264, 0.2666)
 3  ( 0.6124, 0.0000) ( 0.0129,-0.2966)
 4  (-0.0859,-0.3284) ( 0.8898, 0.0000)
```